# $Id: PKGBUILD 72904 2012-06-24 18:16:23Z bgyorgy $
# Maintainer: Balló György <ballogyor+arch at gmail dot com>
# Contributor: Allan McRae <allan@archlinux.org>
# Contributor: Doehni <dohny@gmx.de>

pkgname=gnome-do
pkgver=0.9
pkgrel=2
pkgdesc="Quickly search for many items present on your desktop or the web, and perform useful actions on those items"
arch=('i686' 'x86_64')
url="http://do.davebsd.com/"
license=('GPL3')
depends=('gconf-sharp' 'libgnome-desktop-sharp' 'gnome-keyring-sharp' 'mono-addins' 'notify-sharp' 'hicolor-icon-theme' 'xdg-utils')
makedepends=('intltool' 'libgnome-sharp' 'gio-sharp')
options=('!libtool')
install=$pkgname.install
source=(http://launchpad.net/do/trunk/$pkgver/+download/$pkgname-$pkgver.tar.gz
        gio-gnomedesktop.patch)
md5sums=('d360e56953915d9585c0ee71f433129c'
         'e9ac5e4510c1475d61a20c5d94594d98')

build() {
  cd "$srcdir/$pkgname-$pkgver"

  # https://code.launchpad.net/~ballogy/do/gio-gnomedesktop-port/+merge/95842
  patch -Np1 -i "$srcdir/gio-gnomedesktop.patch"

  autoreconf -fi
  ./configure --prefix=/usr --sysconfdir=/etc \
              --disable-schemas-install \
              --with-gconf-schema-file-dir=/usr/share/gconf/schemas
  make
}

package() {
  cd "$srcdir/$pkgname-$pkgver"

  make DESTDIR="$pkgdir" install
}
